

CORSO DI LAUREA IN INGEGNERIA MECCANICA  
CORSO DI PROGETTAZIONE ASSISTITA DELLE STRUTTURE MECCANICHE

ESERCITAZIONE SULLA CONVERGENZA DELLA SOLUZIONE  
CONVERGENZA AUTOMATICA – EFFETTO DEL GRADO DELLE “SHAPE FUNCTIONS”

Dato l'albero intagliato di fig.1, soggetto a flessione, se ne metta a punto il modello ad EF e si studi la convergenza della tensione equivalente massima in corrispondenza dell'intaglio.




Fig.1. Albero intagliato.

In generale studiare la convergenza della soluzione significa modificare opportunamente il modello in modo tale da ridurre entro limiti prefissati l'errore; può essere utile allo scopo raffinare progressivamente la *mesh* nei punti chiave (*h Convergence*), oppure innalzare il grado delle *shape functions*, lasciando inalterata la *mesh* (*p Convergence*). In ANSYS lo studio di convergenza può essere condotto secondo le due metodologie sopra indicate.

La procedura automatica di *h Convergence* si basa sull'impiego del comando ADAPT, in esecuzione del quale ANSYS conduce una serie di *runs* di soluzione successivi, refinendo di volta in volta la *mesh* con l'obiettivo di ridurre sotto un valore percentuale prestabilito il rapporto fra l'energia associata all'errore e l'energia di deformazione dell'intero modello. La scelta degli elementi da suddividere per affinare la *mesh* viene effettuata in base al valore della *Standard Energy Norm* proprio di ciascun elemento.

La procedura automatica di *p Convergence* richiede invece l'impiego di elementi speciali, appositamente destinati a questo impiego. Per modelli 3D si può usare l'elemento "brick" SOLID147, che fa parte di una famiglia di elementi (*p-elements*) dotati di funzioni di forma con grado variabile. Sostituendo i normali *brick* con il 147, è sufficiente *meshare* una sola volta il volume, lasciando poi che il programma analizzi la convergenza alzando di volta in volta il grado dei polinomi. Tale procedura è regolata dal comando PCONV, con cui si impongono la grandezza da controllare (deformazione, tensione, temperatura,...) e la variazione percentuale massima della stessa, tra due calcoli consecutivi.

Nel post-processor, dal menu "Plot results", è possibile chiedere la rappresentazione del grado delle funzioni di forma e dell'andamento del parametro di convergenza in funzione delle iterazioni.

Si ritiene utile precisare che, contrariamente a quanto avviene per il comando PCONV, il comando ADAPT funziona solo se la *mesh* è *free*. Inoltre, per poter usare tale comando, i vincoli ed i carichi devono necessariamente essere applicati al modello solido (e non al modello ad EF)

Nel seguito viene riportato uno file di comandi utilizzabile per studi di convergenza di tipo "h" e "p".

```

finish
/clear
!
!CORSO DI PROGETTAZIONE ASSISTITA
!
!ESERCITAZIONE DI MODELLAZIONE SOLIDA PARAMETRICA
!E STUDIO DI CONVERGENZA AUTOMATICO
!
!METODO "BOTTOM-UP" - MESHATURA MISTA "FREE"/"MAPPED"
!
!
!DEFINIZIONE DEI PARAMETRI
!
d_mag=20      ![mm]          Diametro maggiore dell'albero
d_min=10      ![mm]          Diametro minore dell'albero
r=3           ![mm]          Raggio di raccordo dell'intaglio
M=10000      ![N*mm]        Momento flettente applicato all'estremità
!
/prep7
!
!CREAZIONE DELLA SEZIONE (MODELLO SIMMETRICO)
!
rectng,0,-r,0,d_min/2+r
rectng,-r,-r-d_mag/2,0,d_mag/2
cyl4,0,d_min/2+r,r,180,,270
asba,1,3
aadd,all
vrotat,all,,,,,kp(0,0,0),kp(-r-d_mag/2,0,0),90
!
!INTRODUZIONE DEGLI ELEMENTI
!
*ask,conv,Studio di convergenza [1=p 2=h],1
!
*if,conv,eq,1,then
et,1,148
conv_1='P'
size1=6
!
*else
et,1,92
conv_1='H'
size1=1
*endif
!
mp,ex,1,210000
!
!PREPARAZIONE DELLA MESH
!
mshkey,0
smrsize,size1
vmesh,all
!
!APPLICAZIONE DEI VINCOLI DI SIMMETRIA
!
asel,,loc,x,-0.0001,0.0001
da,all,symm

```

```

asel,,loc,z,-0.0001,0.0001
da,all,symm
asel,,loc,y,-0.0001,0.0001
da,all,asymm
alls
ksel,,loc,x,-0.0001,0.0001
ksel,r,loc,y,-0.0001,0.0001
ksel,r,loc,z,-0.0001,0.0001
dk,all,uy,0
ksel,all
!
!APPLICAZIONE DEL CARICO SOTTO FORMA DI
!TENSIONE LINEARE IMPOSTA ALLA SEZIONE DI ESTREMITA'
!
sfgrad,pres,0,y,0,-64*M/(3.1415*d_mag**4)
asel,,loc,x,-r-d_mag/2-0.0001,-r-d_mag/2+0.0001
sfa,all,,pres,0
!
allsel
*go,:%conv_1%
/eof

:P
!
!CONVERGENZA DI TIPO "P"
!
!CONTROLLO SULLA CONVERGENZA "P" TRAMITE IL COMANDO PCONV
!
pconv,5,s,eqv,node(0,d_min/2,0)
!
!SOLUZIONE
!
finish
filename,alb_int_P
/title,Albero intagliato sottoposto a flessione (studio di convergenza "P")
!
/solu
solve
!
!POST-PROCESSING
!
/post1
set,1
plnsol,s,eqv
/eof

:H
!
!CONVERGENZA DI TIPO "H"
!
!CONTROLLO SULLA CONVERGENZA "H" TRAMITE IL COMANDO ADAPT
!
finish
/filename,alb_int_H
/title,Albero intagliato sottoposto a flessione (studio di convergenza "H")
!

```

```
/solu  
adapt,10,7.75  
finish  
!  
!POST-PROCESSING  
!  
/post1  
set,1  
plnsol,s,eqv  
/eof
```